

**Timerul T2 (de la microcontrolerul 80C552 - PHILIPS)**

În Fig. 1 se dă schema bloc a Timerului T2.




Fig. 1

**Prezentare:**

Timerul T2 este un timer de 16 biți ce conține 2 registre TMH2 (octet superior) și TML2 (octet inferior). Timerul / numărător de 16 biți poate fi deconectat sau conectat la clock printr-un circuit de prescalare de la una din două surse:  $f_{osc} / 12$  sau de la un semnal extern. Atunci când timerul T2 este configurat ca numărător, circuitul de prescalare primește clock de la un semnal extern pe intrarea T2 (P1.4). Un front crescător la pinul T2 incrementează circuitul de prescalare, iar rata maximă de repetiție este o numărare pe ciclu mașină (1 MHz cu un oscilator de 12 MHz).

Rata maximă de repetiție pentru timerul T2 este de două ori mai mare decât cea a timerului 0 și a timerului 1. T2 (P1.4) este eșantionat la momentul S2P1 și din nou la S5P1 (deci de două ori pe un ciclu mașină). Un front crescător este detectat când T2 este LOW pe durata unui eșantion și HIGH pe durata eșantionului următor. Pentru a asigura detecția frontului crescător, semnalul de intrare trebuie să fie LOW pentru cel puțin 1/2 ciclu și apoi HIGH pentru cel puțin 1/2 ciclu. Dacă este detectat un front crescător înaintea sfârșitului S2P1, timerul va fi incrementat pe durata următorului ciclu; în caz contrar va fi incrementat un ciclu mai târziu. Circuitul prescaler are un factor de prescalare de 1, 2, 4 sau 8 și este șters dacă factorul său de divizare sau sursa de intrare se modifică, ori dacă timerul / numărător este **resetat**.

Timerul T2 poate fi citit "din zbor", dar nu posedă latchuri de citire suplimentare, iar pentru a evita interpretarea eronată a datelor în cazul unei ieșiri de la octetul inferior către octetul superior în timp ce se citește timerul T2 trebuie luate precauții software. Timerul T2 **nu** este încărcabil și este resetat de către semnalul RST sau de un front crescător la intrarea de

semnal RT2 (Reset T2), dacă intrarea a fost activată. RT2 este activat prin intermediul bitului T2ER (SFR - TM2CON.5).

Atunci când octetul cel mai puțin semnificativ al timerului depășește, sau atunci când are loc o depășire pe 16 biți, se poate genera o cerere de întrerupere. Una sau ambele dintre aceste depășiri pot fi programate pentru a solicita o întrerupere. Atunci când octetul inferior (TML2) depășește, se setează flagul T2BO (TM2CON), iar flagul T2OV (TM2IR) este setat când depășește octetul superior (TMH2). De notat că atunci când T2OV este setat, T2BO va fi de asemenea setat. Pentru a activa întreruperile la depășirea octetului inferior trebuie setați biții ET2 (IEN1.7 – activare întrerupere la depășire) și T2IS0 (TM2CON.6 – selecție întrerupere la depășire octet). Bitul T2BO (TM2CON.4) este flagul de **depășire octet** al timerului T2.

Pentru a valida întreruperea la depășirea de 16 biți, trebuie să fie setați biții ET2 (IEN1.7 – activare întrerupere la depășire) și T2IS1 (TM2CON.7 – selecție întrerupere la depășire 16 biți). Pentru a activa atât întreruperile la depășirea de octet cât și la 16 biți, trebuie setați T2IS0 și T2IS1, fiind solicitate două rutine de întrerupere. ***Un test al flagurilor de depășire va indica rutina care va trebui să fie executată.*** Pentru fiecare rutină trebuie șters numai flagul corespunzător depășirii.

Timerul T2 poate fi resetat de un front crescător la RT2 (P1.5) dacă este setat bitul de reset extern al timerului T2 (T2ER) din TM2CON. Acest reset șterge de asemenea și circuitul de prescalare. În modul "mers în gol", timerul numărător și circuitul de prescalare sunt resetate și oprite. Timerul T2 este controlat de registrul de funcții speciale TM2CON.

**Timerul T2. Logica de captură și de comparare.** Timerul T2 este conectat la 4 registre de captură de 16 biți. Un registru de captură poate fi utilizat pentru a "captura" conținutul timerului T2, când are loc o tranziție la pinul său de intrare corespunzător. Un registru de comparare poate fi utilizat pentru a seta, reseta sau inversa prin basculare pinii de ieșire ai **Portului 4** la intervale de timp programabile. Combinația timerului T2 cu logica de captură și comparare este foarte puternică în aplicații ce cuprind mecanisme de rotație, sisteme de injecție pentru automobile, etc.

**Logica de captură.** Cele patru registre de 16 biți la care este conectat timerul T2 sunt: CT0, CT1, CT2 și CT3. Aceste registre sunt încărcate cu conținutul timerului T2, iar o întrerupere este cerută la recepționarea ei la **una din intrările de semnal externe** CT0I, CT1I, CT2I, CT3I. Aceste semnale sunt partajate cu Portul 1. Patru flaguri de întreruperi – corespunzătoare acestor semnale externe de captură sunt în registrul de întreruperi al timerului T2 (registrul de funcții speciale TM2IR). Dacă facilitatea de captură nu este utilizată, aceste intrări pot fi privite ca intrări suplimentare de întreruperi externe.

Folosind registrul de control al capturii CTCON, aceste intrări pot realiza captura pe un front crescător, un front căzător sau fie pe un front crescător și/sau căzător(ambe). Intrările sunt eșantionate pe durata S1P1 a fiecărui ciclu. Când se detectează frontul selectat, conținutul timerului T2 este capturat la sfârșitul ciclului mașină.

**Măsurarea intervalelor de timp folosind registrele de captură.** Atunci când un eveniment extern care se repetă este reprezentat sub forma unui front crescător sau descrescător la unul din cei patru pini de captură, timpul dintre două evenimente poate fi măsurat folosind timerul T2 și un registru de captură. Când un eveniment are loc, conținutul timerului T2 este copiat în registrul de captură corespunzător și se generează o cerere de întrerupere. Rutina de servire a întreruperii poate calcula intervalul de timp dacă se cunoaște conținutul precedent al timerului T2 când a avut loc ultimul eveniment. Cu un oscilator de 12 MHz, timerul T2 poate fi

programat să depășească la fiecare 524 ms. Atunci când intervalul de timp între evenimente este mai scurt decât această valoare, calcularea intervalului de timp este simplă, iar rutina de servire a întreruperii este scurtă. Pentru intervale de timp mai lungi, trebuie utilizată o rutină de extensie a timerului T2, folosind suplimentar octeți din memoria de date – vezi notele de curs!

**Logica de comparare.** De fiecare dată când timerul T2 este incrementat, se compară conținutul celor trei registre de comparare de 16 biți cu noua valoare numărată în timerul T2. Când se constată o egalitate (o coincidență a valorilor), flagul corespunzător al întreruperii este setat în registrul TM2IR la sfârșitul ciclului următor. Când are loc o **coincidență cu CM0**, controlerul setează biții 0 – 5 ai Portului 4, dacă biții corespunzători din registrul de activare a setării **STE** sunt la nivel logic "1".

Când are loc o **coincidență cu registrul CM1**, controlerul resetează biții 0 – 5 din Portul 4 dacă biții corespunzători ai registrului de activare a resetării / basculării RTE sunt la nivel "1" logic (vezi registrul **RTE**). Dacă RTE este "0", atunci P4.n nu este afectat de o împerechere între registrul CM1 sau CM2 și timerul T2.

Când are loc o **coincidență cu registrul CM2**, controlerul basculează biții 6 și 7 din Portul 4 dacă biții corespunzători din **RTE** sunt "1" logic. Latchurile portului pentru biții 6 și 7 nu sunt basculate. Două bistabile suplimentare încarcă ultima operație și de fapt aceste bistabile vor fi basculate. Astfel, dacă operația curentă este "set", următoarea operație va fi "reset", chiar dacă latchul portului este resetat prin software înainte ca operația "reset" să aibă loc. Prima basculare după un reset va seta latchurile portului. Conținutul acestor două bistabile poate fi citit la STE.6 și STE.7 (corespunzând la P4.6, respectiv P4.7). Biții STE.6 și STE.7 sunt numai "citire" (vezi registrul **STE**). Un "1" logic indică faptul că la următoarea basculare se va reseta latchul din port. CM0, CM1 și CM2 sunt resetate de semnalul RST.

Informația modificată în latchul portului apare la pinul portului pe durata S5P1 al ciclului mașină următor celui în care s-a produs coincidența. Dacă portul este modificat prin software, ieșirile se schimbă pe durata S1P1 a ciclului următor. Fiecare bit al portului 4 poate fi setat sau resetat prin software la orice moment de timp. O modificare hardware, rezultând din coincidența cu un comparator ia precedență în raport cu o modificare software în același ciclu. Când rezultatul comparării necesită un "set" și un "reset" în același timp, latchul portului va fi resetat.

## Aplicatie

În cele ce urmează este prezentat un program ce simulează o sirenă de tipul celor montate pe mașinile „salvare”. Dacă îmi doresc ca frecvența semnalului să crească timp de 3 sec. și apoi să scadă tot în 3 secunde, rezultă  $3\text{sec}/100 = 30\text{ms}$  „timpul unei trepte”. „Treptele” le realizez cu ajutorul Timerului T2. Astfel setat, acesta, se „umple” în 30ms, realizându-se „treapta”. Timerul T2 pornește de la 0000H și când ajunge la 7530H (adică 30000us), apare întrerupere dată de Modulul de Comparare 0 – pe acesta încărcându-l inițial cu 7530H. În rutina de întrerupere, prin setarea P4.4 voi **Reseta** Timerul T2, deoarece **extern** am conexiune între P4.4 și P1.5 (RT2). Aceasta este singura posibilitate de a „șterge” Timerul T2, astfel încât acesta să înceapă să numere iar de la valoarea zero. La sfârșitul rutinei de întrerupere, „șterg” P4.4 (nu mai țin Resetat Timerul T2), a.î. acesta poate iar să numere de la noua valoare 0000H.

Din treaptă în treaptă, se modifică frecvența semnalului audio, modificare apărută datorită operațiilor aritmetice „scad” respectiv „adun”. La depășirea (umplerea numărătorului) se setează flagul CMI0. Programul pornește de la o frecvență joasă ce constituie prima treaptă. În registru R2 am încărcat numărul de trepte.

În continuare se dau Regiștrii cu care lucrează Timerul T2:

**Registrul de activare a întreruperilor timerului T2 (IEN1) - adresabil la nivel de bit**

	7	6	5	4	3	2	1	0
<b>IEN1 (E8H)</b>	ET2	ECM2	ECM1	ECM0	ECT3	ECT2	ECT1	ECT0
	(MSB)							(LSB)

Bit	Simbol	Funcție
IEN1.7	ET2	Activare întrerupere (i) depășire timer T2
IEN1.6	ECM2	Activare întrerupere comparator 2 – T2
IEN1.5	ECM1	Activare întrerupere comparator 1 – T2
IEN1.4	ECM0	Activare întrerupere comparator 0 – T2
IEN1.3	ECT3	Activare întrerupere registru de captură 3 – T2
IEN1.2	ECT2	Activare întrerupere registru de captură 2 – T2
IEN1.1	ECT1	Activare întrerupere registru de captură 1 – T2
IEN1.0	ECT0	Activare întrerupere registru de captură 0 – T2

**Registrul de control al timerului T2 (TM2CON)**

	7	6	5	4	3	2	1	0
<b>TM2CON (EAH)</b>	T2IS1	T2IS0	T2ER	T2BO	T2P1	T2P0	T2MS1	T2MS0
	(MSB)							(LSB)

Bit	Simbol	Funcție
TM2CON.7	T2IS1	Selecție întrerupere depășire 16 biți timer T2
TM2CON.6	T2IS0	Selecție întrerupere depășire octet timer T2
TM2CON.5	T2ER	Activare reset extern timer T2. Când acest bit este setat, timerul T2 poate fi resetat de un front crescător la RT2 (P1.5)
TM2CON.4	T2BO	Flag întrerupere depășire octet timer T2
TM2CON.3	T2P1	} Selecție prescalare timer T2
TM2CON.2	T2P0	

T2P1	T2P0	clock timer T2
0	0	Sursa – clock
0	1	Sursa – clock / 2
1	0	Sursa – clock / 4
1	1	Sursa – clock / 8

TM2CON.1	T2MS1	} Selecție mod timer T2
TM2CON.0	T2MS0	

T2MS1	T2MS0	mod selectat
0	0	Timer T2 oprit (off)
0	1	Sursă clock = $f_{osc} / 12$
1	0	Mod test; nu se folosește
1	1	Sursă clock T2 – pinul T2

**Registrul de control al capturii (CTCON)**

	7	6	5	4	3	2	1	0
<b>CTCON (EBH)</b>	CTN3	CTP3	CTN2	CTP2	CTN1	CTP1	CTN0	CTP0
	(MSB)							(LSB)

Bit	Simbol	Funcție
CTCON.7	CTN3	Registrul de captură 3 declanșat de un front căzător la CT3I
CTCON.6	CTP3	Registrul de captură 3 declanșat de un front crescător la CT3I
CTCON.5	CTN2	Registrul de captură 2 declanșat de un front căzător la CT2I
CTCON.4	CTP2	Registrul de captură 2 declanșat de un front crescător la CT2I
CTCON.3	CTN1	Registrul de captură 1 declanșat de un front căzător la CT1I
CTCON.2	CTP1	Registrul de captură 1 declanșat de un front crescător la CT1I
CTCON.1	CTN0	Registrul de captură 0 declanșat de un front căzător la CT0I
CTCON.0	CTP0	Registrul de captură 0 declanșat de un front crescător la CT0I

**Registrul de activare resetare / basculare (RTE)**

	7	6	5	4	3	2	1	0
<b>RTE (EFH)</b>	TP47	TP46	RP45	RP44	RP43	RP42	RP41	RP40
	(MSB)							(LSB)

Bit	Simbol	Funcție
RTE.7	TP47	Dacă este "1", atunci P4.7 basculează atunci când se produce coincidența între CM2 și timerul T2.
RTE.6	TP46	Dacă este "1", atunci P4.6 basculează atunci când se produce coincidența între CM2 și timerul T2.
RTE.5	RP45	Dacă este "1", atunci P4.5 este resetat atunci când se produce coincidența între CM1 și timerul T2.
RTE.4	RP44	Dacă este "1", atunci P4.4 este resetat atunci când se produce coincidența între CM1 și timerul T2.
RTE.3	RP43	Dacă este "1", atunci P4.3 este resetat atunci când se produce coincidența între CM1 și timerul T2.
RTE.2	RP42	Dacă este "1", atunci P4.2 este resetat atunci când se produce coincidența între CM1 și timerul T2.
RTE.1	RP41	Dacă este "1", atunci P4.1 este resetat atunci când se produce coincidența între CM1 și timerul T2.
RTE.0	RP40	Dacă este "1", atunci P4.0 este resetat atunci când se produce coincidența între CM1 și timerul T2.

**Registrul de activare al setării (STE)**

	7	6	5	4	3	2	1	0
<b>STE (EEH)</b>	TG47	TG46	SP45	SP44	SP43	SP42	SP41	SP40
	(MSB)							(LSB)

Bit	Simbol	Funcție
STE.7	TG47	Starea bistabilului
STE.6	TG46	Starea bistabilului
STE.5	SP45	Dacă este "1", atunci P4.5 este setat atunci când se produce coincidența între CM0 și timerul T2.
STE.4	SP44	Dacă este "1", atunci P4.4 este setat atunci când se produce coincidența între CM0 și timerul T2.
STE.3	SP43	Dacă este "1", atunci P4.3 este setat atunci când se produce coincidența între CM0 și timerul T2.
STE.2	SP42	Dacă este "1", atunci P4.2 este setat atunci când se produce coincidența între CM0 și timerul T2.
STE.1	SP41	Dacă este "1", atunci P4.1 este setat atunci când se produce coincidența între CM0 și timerul T2.
STE.0	SP40	Dacă este "1", atunci P4.0 este setat atunci când se produce coincidența între CM0 și timerul T2.

**Registrul flagurilor de întreruperi (TM2IR) - adresabil la nivel de bit**

	7	6	5	4	3	2	1	0
<b>TM2IR (C8H)</b>	T2OV	CM12	CM11	CM10	CT13	CT12	CT11	CT10
	(MSB)							(LSB)

Bit	Simbol	Funcție
TM2IR.7	T2OV	Flagul depășirii la 16 biți ai timerului T2
TM2IR.6	CM12	Flagul întreruperii CM2
TM2IR.5	CM11	Flagul întreruperii CM1
TM2IR.4	CM10	Flagul întreruperii CM0
TM2IR.3	CT13	Flagul întreruperii CT3
TM2IR.2	CT12	Flagul întreruperii CT2
TM2IR.1	CT11	Flagul întreruperii CT1
TM2IR.0	CT10	Flagul întreruperii CT0

**Registrul de priorități ale întreruperilor timerului T2 (IP1) - adresabil la nivel de bit**

	7	6	5	4	3	2	1	0
<b>IP1 (F8H)</b>	PT2	PCM2	PCM1	PCM0	PCT3	PCT2	PCT1	PCT0
	(MSB)							(LSB)

Bit	Simbol	Funcție
IP1.7	PT2	Nivelul de prioritate al întreruperilor la depășire(i) a timerului T2
IP1.6	PCM2	Nivelul de prioritate al întreruperii timer T2 – comparator 2
IP1.5	PCM1	Nivelul de prioritate al întreruperii timer T2 – comparator 1
IP1.4	PCM0	Nivelul de prioritate al întreruperii timer T2 – comparator 0
IP1.3	PCT3	Nivelul de prioritate al întreruperii timer T2 – registru de captură 3
IP1.2	PCT2	Nivelul de prioritate al întreruperii timer T2 – registru de captură 1
IP1.1	PCT1	Nivelul de prioritate al întreruperii timer T2 – registru de captură 2
IP1.0	PCT0	Nivelul de prioritate al întreruperii timer T2 – registru de captură 0

```

$INCLUDE (REG552.INC)           ;foloseste microcontrolerul 80C552
ORG 8000H
JMP START

ORG 805BH                       ;adresa vectorului de intrerupere al Modulului de Comparare 0
JMP INTER                       ;salt la eticheta INTER

START:
MOV IEN0,#80H                   ;activez intreruperile globale EA
MOV IEN1,#90H                   ;activez intrerupere depasire Timer T2
MOV CMH0,#75H                   ;30000 = 7530H - incarc Modulul 0 de comparare
MOV CML0,#30H                   ;
MOV IP1,#90H                    ;stabilesc nivel de prioritate superior
MOV TM2CON,#21H                 ;selectez sursa clock fosc /12 si reset EXTERN Timer T2
MOV R2,#100                     ;registru folosit ca numar de "trepte"
MOV R3,#0                       ;folosit la panta descrescatoare a sunetului
MOV R4,#0                       ;incarc constante pentru operatiuni aritmetice
MOV R5,#03H                     ;incarc constante pentru operatiuni aritmetice
MOV R6,#2                       ;incarc o valoare de start pentru generarea sunetului
MOV R7,#238                     ;incarc o valoare de start pentru generarea sunetului
MOV P4,#00H                     ;"sterg" portul P0

MAIN:
SETB P4.0                       ;rutina de producere a unui semnal audio (vezi lab. 4 si/sau lab.6).
CALL DELAY
CLR P4.0
CALL DELAY
JMP MAIN

DELAY:
MOV 30,R6                       ;intarziere folosita la generarea semnalului audio , cu incarcarea valorilor
MOV 31,R7                       ;din registrii R6 si R7, registri ce se modifica la intrarea in intrerupere
DELAY0: DJNZ 31,DELAY0          ;observati ca am folosit adresele 31 si 30 din RAM !!!
DJNZ 30,DELAY0
RET

INTER:
SETB P4.4                       ;RESET extern Timer T2 - extern am conectat P4.4 = P1.5 (RT2), rezulta TMH2=TML2=00H
DJNZ R2,SCAD                    ;decrementeaza pe R2 si daca nu este zero, sari la "scad"
INC R3                          ;daca R2=0, atunci incrementeaza pe R3
INC R2                          ;incrementeaza pe R2
CJNE R3,#100,ADUN               ;daca R3 nu a ajuns la 50, sari la "adun"
MOV R2,#100                     ;daca R3 este 50, atunci incarca si in R2 valoarea 50
MOV R3,#0                       ;daca am ajuns pana aici, inseamna ca R3=50, acum, sterge pe R3=0.
CLR CM10                        ;sterge flagul Modul 0 de Comparare al Timerului T2 (CM10)
CLR P4.4                        ;dezactivarea Reset extern Timer T2
RETI                            ;revenire din intrerupere

ADUN:
MOV A,R7                        ;muta continutul registrului R7 in Acumulator
ADD A,R5                        ;aduna A cu Carry si R5 , si pune rezultatul in A
MOV R7,A                        ;muta continutul din Acumulator inapoi in registrul R7
MOV A,R6                        ;muta continutul registrului R6 in Acumulator
ADDC A,R4                       ;aduna A cu Carry si R4 , si pune rezultatul in A
MOV R6,A                        ;muta continutul din Acumulator inapoi in registrul R6
CLR C                            ;sterge flagul Carry
CLR CM10                        ;sterge flagul Modul 0 de Comparare al Timerului T2 (CM10)
CLR P4.4                        ;dezactivarea Reset extern Timer T2
RETI                            ;revenire din intrerupere

SCAD:
CLR C                            ;sterge flagul Carry
MOV A,R7                        ;muta continutul registrului R7 in Acumulator
SUBB A,R5                       ;scad A cu Carry si R5 , si pune rezultatul in A
MOV R7,A                        ;muta continutul din Acumulator inapoi in registrul R7
MOV A,R6                        ;muta continutul registrului R6 in Acumulator
SUBB A,R4                       ;scad A cu Carry si R4 , si pune rezultatul in A
MOV R6,A                        ;muta continutul din Acumulator inapoi in registrul R6
CLR CM10                        ;sterge flagul Modul 0 de Comparare al Timerului T2 (CM10)
CLR P4.4                        ;dezactivarea Reset extern Timer T2
RETI                            ;revenire din intrerupere

END                             ;sfarsit de program

```